A note on Bézout’s Theorem

Barry Dayton, 2019

One omission in my book is a proof of Bézout’s theorem. At the time all existing proofs that I knew of used concepts that I have avoided in the book so this omission was intentional. A problem is that my version of intersection multiplicity, which is included in the book specifically for the purposes of the statement of this theorem, is incompatible with these proofs, in particular to the ones by Abyhankar [1] and Fischer [8] that I referenced in the book.

Recently I became aware of a paper by Telen, Mourrain and Van Barel in [Siam J. Matrix Anal Appl, vol 39 no3, pp 1421-1447,(2018)] giving a solution method for two variable polynomial systems which is related to material in Appendix 1, in particular my definition of intersection multiplicity can apply. Although what follows is not a proof of the theorem, it does illustrate the connection between the solutions, including complex, multiple and infinite, and the nullspace of a Sylvester matrix.

My argument involves computing the nullspace of the Sylvester matrix of order $m+n$ of relatively prime polynomials of degree m, n in two different ways and comparing. If you are using the notebook version please initialize global functions and this notebook. There will be some new functions defined here but they will not be added to the global functions page since they are not elsewhere in the book.

In[ ]:=
```
Clear[f1, f2, f3, f4, g1, g2, g3, g4]
```

The nullity of the Sylvester Matrix

Let f, g be the two bivariate polynomials of degrees m, n respectively. Our discussion in section A.3 concludes that if GCD[f,g]=1, that is, if f, g have no common component (factor), then the Sylvester Matrix of order $m+n-1$ or less will have independent rows, that is the row rank will be the number of rows. But since $f\,g$ and $g\,f$ will correspond to two different rows of the Sylvester matrix of order $m+n$ the rank of the rowspace will be only the number of rows minus 1.

The *nullity*, that is the vector space dimension of the null space, will be the number of columns minus the row rank. The number of columns of the Sylvester matrix of order $m+n$ will be the number of monomials of degree less than or equal to $m+n$, calculated in Chapter 2.

In[ ]:=
```
cols = Expand[(n + m + 2) (m + n + 1) / 2]
```

Out[ ]=
$$1+\frac{3\,m}{2}+\frac{m^2}{2}+\frac{3\,n}{2}+m\,n+\frac{n^2}{2}$$

The row rank from the above will be the number of monomials of degree ≤ m plus the number of monomials of degree ≤n minus 1. Thus it will be

In[ ]:=
```
rrank = Expand[((n + 2) (n + 1) + (m + 2) (m + 1)) / 2 - 1]
```

Out[ ]= $1 + \frac{3\,m}{2} + \frac{m^2}{2} + \frac{3\,n}{2} + \frac{n^2}{2}$

Subtracting

In[ ]:=
```
cols - rrank
```

Out[ ]= m n

This actually proves the following Theorem :

Little Bézout Theorem : *Let f,g be bivariate polynomials, of degree m, n respectively, with no common factor. Then the null space of the Sylvester matrix of order m+n of f,g has dimension m n.*

Below we will identify this null space with the intersection points of the two curves, i.e. the solution set of the polynomial system {*f*, *g*} in the complex projective plane. We will work backwards from the solutions so this will not be a proof of Bézout's theorem but an explanation of what is happening.

Case of simple affine solutions

Each affine point will correspond to an *evaluation vector,* or, *e-vector.* We define two simple functions, these will not be listed in Appendix 2 since they are only used here.

In[ ]:=
```
monExps := Table[x ^ p[[1]] y ^ p[[2]], {p, pExps[#]}] &
```

Example:

In[ ]:=
```
monExps[3]
```

Out[ ]= $\{1, x, y, x^2, x\,y, y^2, x^3, x^2\,y, x\,y^2, y^3\}$

Our *evaluation vectors* are now defined for a point p in $\mathbb{C}^2$ and a degree *d* by evaluating each monomial at the same given point.

In[ ]:=
```
eVec[p_, d_] := monExps[d] /. Thread[{x, y} → p];
```

Example :

In[ ]:=
```
e1 = eVec[{3, 2}, 4]
```

Out[ ]= {1, 3, 2, 9, 6, 4, 27, 18, 12, 8, 81, 54, 36, 24, 16}

For simplicity we will use an example from the article TMVB cited above since it has only integer solutions.

In[ ]:=
```
f1 = 7 + 3 x - 6 y - 4 x ^ 2 + 2 x y + 5 y ^ 2;
g1 = -1 - 3 x + 14 y - 2 x ^ 2 + 2 x y - 3 y ^ 2;
```

In[ ]:=
```
sol1 = {x, y} /. NSolve[{f1, g1}]
```
Out[ ]=
```
{{3., 2.}, {2., 1.}, {-1., 0.}, {-2., 3.}}
```

In[ ]:=
```
S1 = sylvesterMatrix[f1, g1, 4, x, y];
NS1 = Transpose[NullSpace[S1]];
NS1 // MatrixForm
```
Out[ ]//MatrixForm=

$$\begin{pmatrix} 390 & 1560 & -1560 & 1 \\ -542 & -2290 & 2032 & -1 \\ -45 & -231 & 150 & 0 \\ 264 & 960 & -1140 & 1 \\ -84 & -348 & 267 & 0 \\ -30 & -180 & 126 & 0 \\ -752 & -3160 & 2758 & -1 \\ -120 & -564 & 465 & 0 \\ -54 & -90 & 180 & 0 \\ 3 & -99 & 81 & 0 \\ 0 & 0 & 0 & 1 \\ -156 & -312 & 663 & 0 \\ 0 & 0 & 234 & 0 \\ 0 & 468 & 0 & 0 \\ 78 & 0 & 0 & 0 \end{pmatrix}$$

Thus the Little Bézout theorem correctly tells us there are 4 columns. Calculating evaluation sequences for the other solutions

In[ ]:=
```
e2 = eVec[{2, 1}, 4]
e3 = eVec[{-1, 0}, 4]
e4 = eVec[{-2, 3}, 4]
```
Out[ ]=
```
{1, 2, 1, 4, 2, 1, 8, 4, 2, 1, 16, 8, 4, 2, 1}
```
Out[ ]=
```
{1, -1, 0, 1, 0, 0, -1, 0, 0, 0, 1, 0, 0, 0, 0}
```
Out[ ]=
```
{1, -2, 3, 4, -6, 9, -8, 12, -18, 27, 16, -24, 36, -54, 81}
```

In[ ]:=
```
S1.e1
S1.e2
S1.e3
S1.e4
```
Out[ ]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```
Out[ ]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```
Out[ ]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```
Out[ ]=
```
{0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

We get 0 in each case! Note one technicality, since Mathematica does not distinguish lists as being rows or columns it writes both rows and columns when given separately as rows. When multiplying it switches rows to columns by the context. So here the Sylvester matrix row is treated as a row while the evaluation sequence row is treated as a column.

So these vectors as columns are in the null space. We already see that e3 is the last column of the nullspace as calculated by Mathematica's **NullSpace** function. But if we look at

In[ ]:=
```
NS1a = Transpose[{e1, e2, e3, e4}];
NS1a // MatrixForm
MatrixRank[NS1a]
S1.NS1a
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 1 & 1 & 1 \\ 3 & 2 & -1 & -2 \\ 2 & 1 & 0 & 3 \\ 9 & 4 & 1 & 4 \\ 6 & 2 & 0 & -6 \\ 4 & 1 & 0 & 9 \\ 27 & 8 & -1 & -8 \\ 18 & 4 & 0 & 12 \\ 12 & 2 & 0 & -18 \\ 8 & 1 & 0 & 27 \\ 81 & 16 & 1 & 16 \\ 54 & 8 & 0 & -24 \\ 36 & 4 & 0 & 36 \\ 24 & 2 & 0 & -54 \\ 16 & 1 & 0 & 81 \end{pmatrix}$$

Out[ ]= 4

Out[ ]=
```
{{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0},
 {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

we see that NS1a is a matrix of rank 4 with columns in the null space of S1 and thus the columns give an alternate basis for the null space of S1.

It is not hard to prove in general that degree d evaluation vectors of solution points of a system $\{f, g\}$ lie in the null space of the Sylvester matrix of f, g of order d. It is harder to prove directly that *these e-vectors of distinct solution points are independent* but will not be proven here.

The case of affine multiple points

If there are affine multiple points, we now show to find additional vectors in the nullspace of the Sylvester matrix. Consider the following Example 2

In[ ]:=
```
f2 = -4 x + 4 x^2 + y^2;
g2 = x^2 + y^2 - 1;
```

The point {1,0} is easily seen to be on both curves.

In[ ]:=
```
intersectionMultiplicity[f2, g2, {1, 0}, 1.*^-12]
```

Out[ ]= 2

From our Dayton-Li-Zeng paper [6] on multiplicity this tells us there will be two local dual differentials at that point, or in other words two independent vectors in the nullspace of the Macaulay matrix at that point. Again we will take the order of the Macaulay matrix to be the sum of the degrees

In[ ]:=
```
M4a = macaulayMatrix[f2, g2, {1, 0}, 4];
 NM4a = Chop[NullSpace[M4a], 1.*^-14]
```

Out[ ]=
```
{{0, 0, 1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}, {1., 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}}
```

However we are interested in the Sylvester matrix which is not dependent on a point. Fortunately there is a transformation which takes null vectors from a Macaulay matrix to the Sylvester matrix. I won't try to fully explain this here, but think of expanding a Taylor series about a point other than zero to get a series based at zero and the binomial coefficients play a large role. This generalizes to the multivariable case and will be discussed in my Space Curve book. In this function *q* is the point and *n* is the order of the Macaulay, Sylvester matrices. This is strictly a matrix operation so no variables are involved

In[ ]:=
```
c2zMat[q_, n_] := Module[{m, Tn, ss, bi, bj, r1, C, s, pow},
  pow[a_, m_] := If[m <= 0, 1, a ^ m];
  s = Length[q];
  Tn = pExps[n];
  ss = Length[Tn];
  ss = Length[Tn];
  C = {};
  Do[bj = Tn[[j]];
   C = Append[C,
     Table[Product[Binomial[Tn[[i]][[k]], bj[[k]]] * pow[q[[k]], (Tn[[i]][[k]] - bj[[k]])], {k, s}],
      {i, ss}]],
   {j, ss}];
  Transpose[C]]
```

In[ ]:=
```
Dimensions[c2zMat[{1, 0}, 4]]
```

Out[ ]=
```
{15, 15}
```

In[ ]:=
```
NM4a0 = c2zMat[{1, 0}, 4].Transpose[NM4a];
NM4a0 // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0. & 1. \\ 0. & 1. \\ 1. & 0. \\ 0. & 1. \\ 1. & 0. \\ 0. & 0. \\ 0. & 1. \\ 1. & 0. \\ 0. & 0. \\ 0. & 0. \\ 0. & 1. \\ 1. & 0. \\ 0. & 0. \\ 0. & 0. \\ 0. & 0. \end{pmatrix}$$

In[ ]:=
```
S4a = sylvesterMatrix[f2, g2, 4, x, y];
S4a.NM4a0
```

Out[ ]=
```
{{0., 0.}, {0., 0.}, {0., 0.}, {0., 0.}, {0., 0.}, {0., 0.},
 {0., 0.}, {0., 0.}, {0., 0.}, {0., 0.}, {0., 0.}, {0., 0.}}
```

So NM5a0 gives 2 independent vectors in the nullspace of S5. Note also that

In[ ]:=
```
u1 = eVec[{1, 0}, 4]
```

Out[ ]=
```
{1, 1, 0, 1, 0, 0, 1, 0, 0, 0, 1, 0, 0, 0, 0}
```

is one of these vectors.

Our calculation of the dimension of the nullspace in the previous section had nothing to do with the multiplicity of the zeros. There are 4 independent vectors since f2 and g2 are of degree 2 We check anyway.

In[ ]:=
```
MatrixRank[NullSpace[S4a]]
```

Out[ ]=
```
4
```

Solving

In[ ]:=
```
sol2 = {x, y} /. NSolve[{f2, g2}]
```

Out[ ]=
```
{{0.333333, 0.942809}, {0.333333, -0.942809}, {1., 0.}, {1., 0.}}
```

Then adding the evaluation vectors for these first two points gives

```
In[ ]:= K = Join[NM4a0, Transpose[{eVec[sol2[[1]], 4], eVec[sol2[[2]], 4]}], 2];
        Dimensions[K]
        MatrixRank[K]
Out[ ]= {15, 4}
Out[ ]= 4
```

```
In[ ]:= Chop[S4a.K]
Out[ ]= {{0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0},
         {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}, {0, 0, 0, 0}}
```

So we can conclude that K is equivalent to the null space of S5a.

As in the earlier section the independence generalizes. Again we will not prove this but assure the reader that it is true.

The case of infinite points

If the rank of the matrix constructed from the affine points, simple and multiple, is not $m\,n$ this suggests infinite points. In this subsection we compute additional vectors in the nullspace from the infinite points. The key is our calculation of infinite points in Chapter 3 of the book. The infinite points of a curve come from solutions of the of maximal form of the equation of the curve. Then the infinite points of the intersection of two curves come from common solutions of the maximal forms of both equations.

I should mention that unless there is some geometrical reason why two curves should share infinite points, that infinite points are rare. In particular if f, g are random curves it is very unlikely that there will be an infinite point in the intersection. Thus our examples are somewhat contrived.

These common forms appear, possibly after multiplication by other polynomials, in the upper degrees of the order $m+n$ Sylvester matrix. For simple, multiplicity 1, crossings one need only look at the highest order. It is easily seen that the number of monomials in two variables of degree exactly $m+n$ is $m+n+1$ so we find the nullspace of the matrix given by the last $m+n+1$ columns of the Sylvester matrix of order $m+n$. Unlike the affine case we handle the multiple case of infinite points together with the simple case but we may need to look in a lower order as well.

A simple example is the intersection of two rational function curves $y=\frac{1}{x-2}$, $y=\frac{-1}{x+2}$

```
In[ ]:= f3 = Expand[y (x - 2) - 1]
        g3 = Expand[y (x + 2) + 1]
Out[ ]= -1 - 2 y + x y
Out[ ]= 1 + 2 y + x y
```

In[ ]:=
```
ContourPlot[{f3 == 0, g3 == 0}, {x, -5, 5}, {y, -10, 10}, ImageSize → Small]
```

Out[ ]=


In[ ]:=
```
NSolve[{f3, g3}]
```

Out[ ]= {{x → 0., y → -0.5}}

These are both quadratics so we expect 4 intersection points, hence 3 must be infinite.

In[ ]:=
```
infiniteRealPoints[f3, x, y]
infiniteRealPoints[g3, x, y]
```

Out[ ]= {{-1.40119, 0., 0}, {0., -1.35178, 0}}

Out[ ]= {{3.04916, 0., 0}, {0., 3.20646, 0}}

Two are found, essentially {1, 0, 0}, {0, 1, 0}, but one must be multiple. One can check using the method of Chapter 8 but from the plot it appears that {1, 0, 0} is the culprit, which is actually the case. Calculating the Sylvester matrix and its nullspace gives

In[ ]:=
```
S3 = sylvesterMatrix[f3, g3, 4 , x, y];
Dimensions[S3]
ns3 = Transpose[NullSpace[S3]];
ns3 // MatrixForm
```

Out[ ]= {12, 15}

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & -8 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 4 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & -2 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 \end{pmatrix}$$

with 4 columns as expected. But finding the nullspace of the last 5 columns of S3, the degree 4 part, gives only

In[ ]:=
```
nsinf = Transpose[NullSpace[Take[S3, All, -5]]];
nsinf // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 1 \\ 0 & 0 \\ 0 & 0 \\ 0 & 0 \\ 1 & 0 \end{pmatrix}$$

So we try including the degree 3 part which has 4 columns.

In[ ]:=
```
nsinf = Transpose[NullSpace[Take[S3, All, -9]]];
nsinf // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \\ 1 & 0 & 0 \end{pmatrix}$$

This has the desired rank 3 corresponding to 3 infinite points by multiplicity. Since the actual Sylvester matrix S11 has 15 columns we prepend a 6×3 matrix of zeros

In[ ]:=
```
infpart = Join[Table[0, {i, 6}, {j, 3}], nsinf];
Dimensions[infpart]
```

Out[ ]= {15, 3}

Finally we can add the evaluation vector corresponding to the affine solution {0, −.5} as a column

In[ ]:=
```
ev1 = Partition[eVec[{0, -.5}, 4], 1];
nullMat = Join[ev1, infpart, 2];
nullMat // MatrixForm
```

Out[ ]//MatrixForm=

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ -0.5 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \\ 0. & 0 & 0 & 0 \\ 0.25 & 0 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0. & 0 & 0 & 0 \\ 0. & 0 & 0 & 0 \\ -0.125 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0. & 0 & 0 & 0 \\ 0. & 0 & 0 & 0 \\ 0. & 0 & 0 & 0 \\ 0.0625 & 1 & 0 & 0 \end{pmatrix}$$

We see

In[ ]:=
```
MatrixRank[nullMat]
S3.nullMat
```

Out[ ]= 4

Out[ ]=
```
{{0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.},
 {0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.},
 {0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.}, {0., 0., 0., 0.}}
```

so nullMat gives a basis for the nullspace of S11 once again explaining this nullspace in terms of the intersection points of the two curves.

Again I assert, but will not prove, that this example is typical.

One Final example

We now do one final example collecting these ideas, bringing in complex solutions.

In[ ]:=
```
f4 = -7 - 4 x - 4 x^2 + x^3 - 5 x y + 4 x^2 y + 9 y^2 - 7 x y^2 - 10 y^3;
g4 = 1 - 3 x - 7 x^2 + x y - 3 x^2 y + x^3 y + 6 x y^2 + 4 x^2 y^2 + 3 y^3 - 7 x y^3 - 10 y^4;
```

Our Sylvester matrix will be of order 3 + 4 = 7. We get the expected results as to size of the Sylvester matrix and its nullspace.

In[ ]:=
```
S4 = sylvesterMatrix[f4, g4, 7, x, y];
Dimensions[S4]
NS4 = Transpose[NullSpace[S4]];
Dimensions[NS4]
```

Out[ ]= {25, 36}

Out[ ]= {36, 12}

A plot of this is

In[ ]:=
```
ContourPlot[{f4 == 0, g4 == 0}, {x, -8, 8}, {y, -8, 8}, ImageSize → Small]
```

Out[ ]=


In[ ]:=
```
sol4 = {x, y} /. NSolve[{f4, g4}]
Length[sol4]
```

Out[ ]= {{34.9533, -6.47384}, {4.64517, 1.4483},
{0.305091 - 0.862142 i, 0.332544 - 0.825219 i}, {0.305091 + 0.862142 i, 0.332544 + 0.825219 i},
{-0.157108 - 0.499093 i, -0.57775 + 0.0373497 i},
{-0.157108 + 0.499093 i, -0.57775 - 0.0373497 i},
{1.43034, -1.2797}, {-0.503689 - 0.198654 i, 0.658593 - 0.214583 i},
{-0.503689 + 0.198654 i, 0.658593 + 0.214583 i}}

Out[ ]= 9

We see that there are 9 affine solutions, 3 of which are real. Two of the real solutions are shown in the plot. There appear to be 3 infinite solutions. We now attempt to explain the nullspace NS4 by the solutions.

In[ ]:=
```
Naff = Transpose[Table[eVec[sol4[[j]], 7], {j, 9}]];
Dimensions[Naff]
Ninf = Join[Table[0, {i, 28}, {j, 3}], Transpose[NullSpace[Take[S4, All, -8]]]];
Dimensions[Ninf]
```

Out[ ]= {36, 9}

Out[ ]= {36, 3}

In[ ]:=
```
NS4all = Join[Naff, Ninf, 2];
MatrixRank[NS4all]
SingularValueList[NS4all]
terr = S4.NS4all;
Print["Total error ", Total[Abs /@ Flatten[S4.NS4all]]]
```

Out[ ]= 12

Out[ ]= $\{6.48891\times10^{10}$, 25 060.3, 178.362, 29.049, 6.51637, 3.7695, 3.52223, 2.22062, 1.49799, 1.08137, 0.956131, 0.745054$\}$

Total error 0.000232134

We checked the rank calculation with the singular value list, the columns are clearly independent. The total error of all 300 entries of the product S4.NS4all is quite small, suggesting that the columns NS4all give a reasonable approximation of the nullspace of S4.

Conclusion

From the discussion above in all the examples we were able to completely describe the null space of the Sylvester matrix of order $m + n$ in terms of the calculated solutions. While this is not an actual proof of Bézout's theorem this should give some insight as to why it works. A complete proof would need to theoretically prove independence of these columns and also show existence of these intersections from the nullspace. The cited paper of Telen, Mourrain and Van Barel above does show this existence for at least the affine solutions. On the other hand, working backwards from Bézout's theorem we see that our claim of the existence and independence of this basis for the nullspace of the Sylvester matrix is supported.

As a bonus, we see that to calculate the intersection multiplicity we only need to calculate one Macaulay matrix, that of order $m + n$ and find the dimension of its nullspace. We propose a new algorithm

In[ ]:=
```
intersectionMultiplicity2[f_, g_, p_, x_, y_] := Module[{m, n, M, c, svdl},
  m = tDeg[f, x, y];
  n = tDeg[g, x, y];
  M = macaulayMatrix[f, g, p, n + m];
  c = Dimensions[M][[2]];
  svdl = SingularValueList[M, Tolerance → 1.*^-12];
  c - Length[svdl]]
```

In[ ]:=
```
{t1, i1} = Timing[intersectionMultiplicity2[f2, g2, {1, 0}, x, y]]
{t2, i2} = Timing[intersectionMultiplicity[f2, g2, {1, 0}, 1.*^-10]]
```

Out[ ]= {0.006049, 2}

Out[ ]= {0.001282, 2}

Unsurprisingly, the new algorithm may not be faster when the multiplicity is small, but look what happens when the multiplicity (and depth) is large.

```
In[ ]:= {t3, i3} = Timing[intersectionMultiplicity2[x - y + x^8, x - y + y^8, {0, 0}, x, y]]
{t4, i4} = Timing[intersectionMultiplicity[x - y + x^8, x - y + y^8, {0, 0}, 1*^-10]]

Out[ ]= {0.419866, 15}

Out[ ]= {2.81598, 15}
```

I suspect that this bound $m + n$ of the depth (see our paper [6]) of a multiple point of the solution of a 2×2 polynomial system generalizes to N×N systems. That is, if the degrees of the N equations are $d_{1,} \ldots, \; d_N$ then the bound on the depth is $D = d_1 + \cdots + d_N$. Unfortunately even this bound is too large to be of much help in calculating intersection multiplicity for $N > 2$ so such a result would only be of theoretical interest.

Bézout's theorem does extend to the multivariable case of a system of r polynomial equations in $n \leq r$ variables. Classically for $n = r$ it says that that if the system is *zero-dimensional,* that is has a finite number of solutions, then the number of complex projective solutions is $d_1 * d_2 * \cdots * d_n$where, as above, the d_j are the degrees of the equations. This is a difficult theorem of algebraic geometry. However there is strong evidence that in the general zero-dimensional case the dimension of nullspace of a multivariable Sylvester matrix of order $m \geq D$ for D as in the paragraph above, counts the number complex projective solutions by multiplicity generalizing the *little Bézout theorem* above. For further information see my **Summary.pdf** on my space curve page of my website.